C.

BLOCK DIAGRAM OF BEST FIT PROGRAM

B. Ronne,
TC Division, CERN



7863/p

MAIN PROGRAM FOR BEST FIT OF LAMBDA DECAY (PAGE 103-112)

```
      DIMENSION EMO(4),EM(4),C(4),F(3),B(12),A(3),GI(16),GMFI(16)
      COMMON NF,NM,NX,F,B,A,GI,C,DX,GMFI,GXI,CHI
C               READ GENERAL DATA
      READ INPUT TAPE 1,1,NF,NM,NX,FLIM
    1 FORMAT (3I10, F10.5)
C               READ ONE EVENT
    2 READ INPUT TAPE 1,3,NEVENT,(EMO(I),I=1,4),(GI(I),I=1,16)
    3 FORMAT (I10/(4F10.5))
      IF(EMO(1)) 5,4,5
    4 CALL EXIT
C               APPROXIMATIVE VALUE OF X
    5 X = EMO(1)*COSF(EMO(2)) + EMO(3)*COSF(EMO(4))
C               STARTING VALUES OF ITERATION PROCESS
      NSTEP = 0
      DO 6 I=1,4
    6 C(I) = 0.
      DX = 0.
C               NEW SET OF VARIABLES
    7 DO 8 I=1,4
    8 EM(I) = EMO(I) + C(I)
      X = X + DX
C               CONSTRAINT FUNCTIONS
      F(1) = X - EM(1)*COSF(EM(2)) - EM(3)*COSF(EM(4))
      F(2) = EM(1)*SINF(EM(2)) - EM(3)*SINF(EM(4))
      F(3) = SQRTF(X**2+1.1154**2) - SQRTF(EM(1)**2+0.9382**2) -
     1       SQRTF(EM(3)**2+0.1396**2)
      IF(ABSF(F(1))+ABSF(F(2))+ABSF(F(3)) - FLIM) 11,9,9
    9 NSTEP = NSTEP + 1
      IF(NSTEP - 10) 10,10,13
C               DERIVATIVE MATRICES B AND A
   10 B(1) = -COSF(EM(2))
      B(2) = EM(1)*SINF(EM(2))
      B(3) = -COSF(EM(4))
      B(4) = EM(3)*SINF(EM(4))
      B(5) = SINF(EM(2))
      B(6) = EM(1)*COSF(EM(2))
      B(7) = -SINF(EM(4))
      B(8) = -EM(3)*COSF(EM(4))
      B(9) = -EM(1)/SQRTF(EM(1)**2 + 0.9382**2)
      B(10)= 0.
      B(11)= -EM(3)/SQRTF(EM(3)**2 + 0.1396**2)
      B(12)= 0.
      A(1) = 1.
      A(2) = 0.
      A(3) = X/SQRTF(X**2 + 1.1154**2)
C               NEW CORRECTIONS C AND DX
      CALL BOECK(0)
      GO TO 7
C               ERROR MATRICES
   11 CALL BOECK(1)
      WRITE OUTPUT TAPE 2,12,NEVENT,(EMO(I),I=1,4),(GI(I),I=1,16),
     1 (EM(I),I=1,4),(GMFI(I),I=1,16),(B(I),I=1,4),X,GXI,CHI,NSTEP
   12 FORMAT (1H1 7H NEVENT I5//4H EMO / 4F10.5//3H GI/4F10.5/4F10.5/
     1 4F10.5/4F10.5//3H EM /4F10.5//5H GMFI /4F10.5/4F10.5/4F10.5/
     2 4F10.5//2H B /4F10.5//2H X/ 1F10.5//4H GXI/1F10.5//4H CHI/1F10.5
     3 // 6H NSTEP /I10)
      GO TO 2
   13 WRITE OUTPUT TAPE 2,14, NEVENT
   14 FORMAT (1H1 I10/ 12H NO SOLUTION )
      GO TO 2
      END
```

7863/p

DATA FOR LAMBDA DECAY PROGRAM

3	4	1	0.00001

11

0.9	0.1	0.1	1.0
0.01	0.	0.	0.
0.	0.0003	0.	0.
0.	0.	0.0004	0.
0.	0.	0.	0.003

----

6 blank cards

RESULTS FROM THE FIT

NEVENT 11

EMO

0.90000	0.10000	0.10000	1.00000

GI

0.01000	0.	0.	0.
0.	0.00030	0.	0.
0.	0.	0.00040	0.
0.	0.	0.	0.00030

EM

0.80672	0.09432	0.09042	0.99787

GMFI

0.00234	-0.00065	-0.00036	-0.00014
-0.00065	0.00019	0.00011	-0.00000
-0.00036	0.00011	0.00006	-0.00003
-0.00014	-0.00000	-0.00003	0.00030

B

0.00220	-0.00060	-0.00033	-0.00018

X

0.85215

GXI

0.00207

CHI

1.22231

NSTEP

2

7863/p

EXERCISE IN WRITING A BEST FIT PROGRAM.

Problem: Test if two gamma rays, which have converted to pairs, come from the same origin, and find the coordinates of this origin.

We assume the apex of each pair to be so well known that we can treat the coordinates as fixed numbers and have, therefore, for each gamma ray, two measured quantities; dipangle (λ) and azimuthangle (φ) with errors.

Notations:



Constraint functions

$$\frac{x_o - x_A}{l_A} = \frac{y_o - y_A}{m_A} = \frac{z_o - z_A}{n_A}$$

$$\frac{x_o - x_B}{l_B} = \frac{y_o - y_B}{m_B} = \frac{z_o - z_B}{n_B}$$

with

$$l_i = \cos\lambda_i \cos\varphi_i$$

$$m_i = \cos\lambda_i \sin\varphi_i \qquad i = A,B$$

$$n_i = \sin\lambda_i$$

We write the constraint functions in the following way:

$$\begin{cases} F(1) = m_A(x_o - x_A) - l_A(y_o - y_A) = 0 \\ F(2) = n_A(x_o - x_A) - l_A(z_o - z_A) = 0 \\ F(3) = m_B(x_o - x_B) - l_B(y_o - y_B) = 0 \\ F(4) = n_B(x_o - x_B) - l_B(z_o - z_B) = 0 \end{cases}$$

We have 4 equations, 4 measured variables (λ_A, φ_A, λ_B, φ_B) and 3 unknown ones (x_o, y_o, z_o). This gives a 1C-fit.

Approximative values of the unknown variables.

From the constraint equations we can solve the unknowns in different ways (with different results). One solution is:

$$\begin{cases} x_o = \dfrac{l_A l_B (y_B - y_A) + m_A l_B x_A - m_B l_A x_B}{m_A l_B - m_B l_A} \\[2ex] y_o = \dfrac{m_A m_B (x_A - x_B) + m_A l_B y_B - m_B l_A y_A}{m_A l_B - m_B l_A} \\[2ex] z_o = \dfrac{n_A n_B (x_A - x_B) + n_A l_B z_B - n_B l_A z_A}{n_A l_B - n_B l_A} \end{cases}$$

Derivative matrices B and A.

$$B = \begin{vmatrix} \dfrac{dF(1)}{d\lambda_A} & \dfrac{dF(1)}{d\varphi_A} & \dfrac{dF(1)}{d\lambda_B} & \dfrac{dF(1)}{d\varphi_B} \\ \vdots & & & \\ \dfrac{dF(4)}{d\lambda_A} & \cdots & \cdots & \dfrac{dF(4)}{d\varphi_B} \end{vmatrix}$$

$$B(1,1) = -(x_o-x_A)\sin\lambda_A\sin\varphi_A + (y_o-y_A)\sin\lambda_A\cos\varphi_A$$

$$B(1,2) = (x_o-x_A)\cos\lambda_A\cos\varphi_A + (y_o-y_A)\cos\lambda_A\sin\varphi_A$$

$$B(2,1) = (x_o-x_A)\cos\lambda_A + (z_o-z_A)\sin\lambda_A\cos\varphi_A$$

$$B(2,2) = (z_o-z_A)\cos\lambda_A\sin\varphi_A$$

$$\left.\begin{aligned} B(3,3) &= B(1,1) \\ B(3,4) &= B(1,2) \\ B(4,3) &= B(2,1) \\ B(4,4) &= B(2,2) \end{aligned}\right\} \text{ with B instead of A}$$

The other components are 0.

$$A = \begin{vmatrix} \frac{dF(1)}{dx_o} & \frac{dF(1)}{dy_o} & \frac{dF(1)}{dz_o} \\ \frac{dF(2)}{dx_o} & \frac{dF(2)}{dy_o} & \frac{dF(2)}{dz_o} \\ \frac{dF(3)}{dx_o} & \frac{dF(3)}{dy_o} & \frac{dF(3)}{dz_o} \\ \frac{dF(4)}{dx_o} & \frac{dF(4)}{dy_o} & \frac{dF(4)}{dz_o} \end{vmatrix} = \begin{vmatrix} m_A & -l_A & 0 \\ n_A & 0 & -l_A \\ m_B & -l_B & 0 \\ n_B & 0 & -l_B \end{vmatrix}$$

<u>Notations to be used in the program.</u>

$x_A, y_A, z_A, l_A, m_A, n_A \Rightarrow$ AX,AY,AZ,AL,AM,AN

and the same notation for B-quantities.

λ_A = EMO(1)
φ_A = EMO(2)
λ_B = EMO(3)
φ_B = EMO(4)

} Measured quantities

Fitted quantities are denoted EM(I), I=1,4.

x_o = X(1)

y_o = X(2)

z_o = X(3)

Other notations as in SUBROUTINE BOECK.

BEST FIT PROGRAM TO FIND THE ORIGIN OF TWO GAMMA RAYS

```
      DIMENSION EMO(4),EM(4),C(4),F(4),X(3),DX(3),B(16),A(12),GI(16),
     1      GMFI(16),GXI(9)
      COMMON NF,NM,NX,F,B,A,GI,C,DX,GMFI,GXI,CHI
C                READ GENERAL DATA
      READ INPUT TAPE 1,1,NF,NM,NX,FLIM
    1 FORMAT (3I10,F10.5)
C                READ ONE EVENT
    2 READ INPUT TAPE 1,3,NEVENT,(EMO(I),I=1,4),(GI(I),I=1,16),AX,AY,
     1      AZ,BX,BY,BZ
    3 FORMAT (I10/(4F10.5))
      IF(EMO(1)) 5,4,5
    4 CALL EXIT
C                APPROXIMATIVE VALUES OF X(I)
    5 CALL DIRCOS (EMO,AL,AM,AN,BL,BM,BN)
      X(1) = (AL*BL*(BY-AY) + AM*BL*AX - BM*AL*BX)/(AM*BL - BM*AL)
      X(2) = (AM*BM*(AX-BX) + AM*BL*BY - BM*AL*AY)/(AM*BL - BM*AL)
      X(3) = (AN*BN*(AX-BX) + AN*BL*BZ - BN*AL*AZ)/(AN*BL - BN*AL)
C                STARTING VALUES OF ITERATION PROCESS
      NSTEP = 0
      DO 6 I=1,4
    6 C(I) = 0.
      DO 7 I=1,3
    7 DX(I) = 0.
C                NEW SET OF VARIABLES
    8 DO 9 I=1,4
    9 EM(I) = EMO(I) + C(I)
      DO 10 I=1,3
   10 X(I) = X(I) + DX(I)
C                CONSTRAINT FUNCTIONS
      CALL DIRCOS (EM,AL,AM,AN,BL,BM,BN)
      F(1) = AM*(X(1) - AX) - AL*(X(2) - AY)
      F(2) = AN*(X(1) - AX) - AL*(X(3) - AZ)
      F(3) = BM*(X(1) - BX) - BL*(X(2) - BY)
      F(4) = BN*(X(1) - BX) - BL*(X(3) - BZ)
      IF(ABSF(F(1))+ABSF(F(2))+ABSF(F(3))+ABSF(F(4))-FLIM) 15,11,11
   11 NSTEP = NSTEP + 1
      IF(NSTEP - 10) 12,12,17
C                DERIVATIVE MATRICES B AND A
   12 DO 13 I=1,16
   13 B(I) = 0.
      B(1) = -(X(1) - AX)*SINF(EM(1))*SINF(EM(2)) + (X(2) - AY)*
     1      SINF(EM(1))*COSF(EM(2))
      B(2) =  (X(1) - AX)*COSF(EM(1))*COSF(EM(2)) + (X(2) - AY)*
     1      COSF(EM(1))*SINF(EM(2))
      B(5) =  (X(1) - AX)*COSF(EM(1)) + (X(3) - AZ)*SINF(EM(1))*
     1      COSF(EM(2))
      B(6) =  (X(3) - AZ)*COSF(EM(1))*SINF(EM(2))
      B(11)= -(X(1) - BX)*SINF(EM(3))*SINF(EM(4)) + (X(2) - BY)*
     1      SINF(EM(3))*COSF(EM(4))
      B(12)=  (X(1) - BX)*COSF(EM(3))*COSF(EM(4)) + (X(2) - BY)*
     1      COSF(EM(3))*SINF(EM(4))
      B(15)=  (X(1) - BX)*COSF(EM(3)) + (X(3) - BZ)*SINF(EM(3))*
     1      COSF(EM(4))
      B(16)=  (X(3) - BZ)*COSF(EM(3))*SINF(EM(4))
```

```
      DO 14 I=1,12
   14 A(I) = 0.
      A(1) =  AM
      A(2) = -AL
      A(4) =  AN
      A(6) = -AL
      A(7) =  BM
      A(8) = -BL
      A(10)=  BN
      A(12)= -BL
C                 NEW CORRECTIONS C AND DX
      CALL BOECK(0)
      GO TO 8
C                 ERROR MATRICES
   15 CALL BOECK(1)
      WRITE OUTPUT TAPE 2,16,NEVENT,(EMO(I),I=1,4),(GI(I),I=1,16),
     1      (EM(I),I=1,4),(GMFI(I),I=1,16),(X(I),I=1,3),(GXI(I),I=1,9),
     2      (B(I),I=1,12),CHI,NSTEP
   16 FORMAT (1H1 7H NEVENT I5//4H EMO/4F10.5//3H GI/4F10.5/4F10.5/
     1      4F10.5/4F10.5//3H EM/4F10.5//5H GMFI/4F10.5/4F10.5/4F10.5/
     2      4F10.5//2H X/3F10.5//4H GXI/3F10.5/3F10.5/3F10.5//2H B/
     3      3F10.5/3F10.5/3F10.5/3F10.5//4H CHI/F10.5//6H NSTEP/I5)
      GO TO 2
   17 WRITE OUTPUT TAPE 2,18,NEVENT
   18 FORMAT (1H1 I10// 12H NO SOLUTION)
      GO TO 2
      END


C  SUBROUTINE WHICH GIVES DIRECTION COSINES OF THE TWO GAMMAS

      SUBROUTINE DIRCOS (EM,AL,AM,AN,BL,BM,BN)
      DIMENSION EM(4)
      AL = COSF(EM(1))*COSF(EM(2))
      AM = COSF(EM(1))*SINF(EM(2))
      AN = SINF(EM(1))
      BL = COSF(EM(3))*COSF(EM(4))
      BM = COSF(EM(3))*SINF(EM(4))
      BN = SINF(EM(3))
      RETURN
      END
```

DATA FOR GAMMA PROGRAM

4	4	3	0.001

123

0.75	1.10	-0.40	-0.49
0.002	0.	0.	0.
0.	0.003	0.	0.
0.	0.	0.004	0.
0.	0.	0.	0.005
3.	4.	4.	4.
1.	1.		

----

8 blank cards

RESULTS FROM THE FIT

NEVENT 123

EMO

0.75000	1.10000	-0.40000	-0.49000

GI

0.00200	0.	0.	0.
0.	0.00300	0.	0.
0.	0.	0.00400	0.
0.	0.	0.	0.00500

EM

0.75190	1.09941	-0.40247	-0.48759

GMFI

0.00122	0.00024	-0.00102	-0.00099
0.00024	0.00292	-0.00032	0.00031
0.00102	-0.00032	0.00268	0.00129
-0.00099	0.00031	0.00129	0.00373

X

2.00902	2.05580	1.95936

GXI

0.01347	0.00103	0.00032
0.00103	0.01940	0.00953
0.00032	0.00953	0.01996

B

0.00148	0.00175	-0.00291
0.00530	-0.00362	-0.00185
-0.00193	-0.00227	-0.00684
-0.00322	-0.00782	-0.00380

CHI

0.00461

NSTEP

1

7863/p